
Simulating Emergy and Materials in Hierarchical Steps

Howard T. Odum

ABSTRACT

Understanding the self organization of systems of nature and environment requires synthesis of several principles for the hierarchy of territories, energy transformations, the coupling of material cycles, the formation of centers, and the concentration of emergy and matter. These relationships were combined in a BASIC computer program HIERSTEP.bas that simulates the properties of hierarchical steps of a system in steady state, each level calculated from inputs from the next lower, less concentrated level. Simulations are useful for controlled experimentation on the effects of changes on hierarchical characteristics. Results are either printed as numbers or plotted as heights on a bar graph on logarithmic vertical scale. The output plot shows steps increasing or decreasing with hierarchical level from left to right. Parameters calculated for each level in the hierarchy include: area of territory, number of centers, area of centers, area of each center, emergy flow, empower, coupled material flow, areal empower density, areal mass flux density, transformity of flow, emergy per gram of flow, storage turnover times, emergy stored, and matter stored. Questions raised by this exercise are used to discuss hypotheses for the organization of ecosystems, landscapes, and economies. The process of combining the relationships into one program identifies parameters that need to be evaluated in real systems. For example, a material-emergy function relates the amount of a material cycle that is coupled to available emergy in a transformation as a function of the level in emergy hierarchy.

INTRODUCTION

Study of the hierarchy of nature has a long history with many approaches and concepts (Allen and Starr, 1982; Clarke, 1946; Mandelbrot, 1982; Margalef, 1958; Mesarovic et al., 1970; Odum, 1982; Patee, 1973; Steele, 1976; Weiss, 1971; Whyte and Wilson, 1969; Zipf, 1941), to mention a few. Hierarchy involves the relationship of scales, energy transformations, material concentrations and cycles, spatial organization of territories and centers, controls, system reinforcements, and pulsing impacts. Berry et al. (1976) reviewed the concepts of hierarchy in human settlements that started in the tenth century with later measures proposed by Zipf (1941) and Doxiadis (1977). Emergy and transformity and related indices used to characterize energy and material hierarchies were summarized by this author (1996). In this paper many of these properties are combined into a computer simulation program that relates each to the others, directly or indirectly. Running the program with various changes in parameters and mechanisms is an experimental research tool for understanding the steps of hierarchy. Procedures used in BASIC simulation are given in detail elsewhere (Odum and Odum 2000).

Program Development

The various concepts of energy hierarchy and its coupling to one material cycle are summarized in Table 1. These relationships were combined in a BASIC language computer simulation program (HIERSTEP.bas) that plots bar graphs for each variable in successive levels starting with the values of the




Figure 1. Diagram of four levels of energy hierarchy programmed in the BASIC model HIERSTEP.bas. (a) Spatial sketch of the centers and pathways converging toward centers; (b) energy systems diagram with energy flows transforming, feeding back reinforcement actions, and dispersing; (c) material circulation added to energy systems diagram to show processes of concentration, recycle and dispersal. Shaded symbols represent material storage in a transformation unit.

Table 1
Relationships introduced into the hierarchical steps program
HIERSTEP.bas in Table 3
* = new parameters for data evaluation.

Calculations were made with energy flows and materials circulating through the system at a steady state (flows constant).

Hierarchical level (L) was started at 1 and increased by step ($S = 1$) in calculating the properties of the next level.

Area of a territory at each level at was calculated for the first level as the total area ($A = 1 \times 10^9$) divided by the number of territories ($nc = 1000$). There was one center for each territory. Area of each center A_{ec} was a set fraction ($Ca = 0.05$)* and area of all these centers times their number: $Ac = (A_{ec})(nc)$.

For the next levels the number of territorial centers converging outputs was ($Cr = 6$)* so that the next level was increased in area with each step by 6 and the number of centers and territories divided by 6.

The area of all the centers at a higher level Ac^* was the fraction ($Sf = 0.3$)* of the area of all the centers at the lower level. $Ac = (Sf)(Ac)$.

The number of centers nc^* is the total area divided by the area of each territory at that level. ($nc = A/at$)

Constant empower (emp) was flowing through the whole area. ($emp = 1 \times 10^{20}$).

Constant low quality inflowing energy Je was spread over the whole area.

At each energy transformation sending flow to the next level, fraction of energy transformed was the efficiency ($Eff = 0.1$).

Transformity of the flows was the empower (emp) divided by the energy flow (Je).

The total mass of a material M^* at the low level was spread over the whole area ($M = 1 \times 10^7$).

Flow of the mass from the background level was coupled to the flow of energy Je according to the ratio me and $J_m = (me)(Je)$. But the coupling me^* was increased with the level changing with the coefficient f_{mc} and level L . Thus, $me = (me)(f_{mc})(L)$. These properties represent the tendencies of different materials to be coupled to available energy transformations according to their quality.

Empower density is the empower divided by the area of centers at that level. ($emp_d = emp/ac$)

Energy per mass for the flows is the empower divided by the mass flow ($emp_g = emp/J_m$)

The concentration of material flow in centers (material flux density)* is the material flow at a level divided by the area of its centers ($m_{fd} = J_m/ac$).

Steady state storages were calculated from the flows times the turnover times which increase with scale (level).


Turnover time of the storages in each center R was increased by a factor R_f^* at the next hierarchical level where $R_f = 10$.

Storage of materials is the flow of materials at a level times its time for accumulation (turnover time) and $ms = (J_m)(R)$.

Storage of energy is the empower at a level times its turnover time $ems = (emp)(R)$.

Areal concentration of materials mc^* is the material storage in centers at a level divided by the area of those centers ($mc = ms/ac$).

(a) Spatial Properties on Two Levels



(b) Properties of One Center




Figure 2. Diagrams identifying some parameters and abbreviations used in the program HIERSTEP.bas. (a) Spatial properties; (b) concentration of materials and storing of available energy in a transformation center.

background environment. Figure 1a is a spatial sketch of different levels of territory and territorial centers, each concentrating and transforming flows of energy, empower, and materials which are passed then to the next higher, more concentrated level. Figure 1b is the energy systems diagram of the background and four levels of concentrating. The hierarchical characteristics that were calculated and plotted are listed in Table 2 and some are illustrated in Figure 2. Figure 3 has the bar graphs generated by the computer simulation with the initial calibration.

The process of combining the various relationships, required new attention to the equations for some relationships, forcing some new parameters to be defined. As a general synthesis of concepts, the model identifies measurements (Table 2) that need to be calculated from data of various kinds of systems to further test theories about hierarchical relationships. These parameters, necessarily introduced to make the principles mutually consistent, are marked with *.

Running the model with different numerical values for parameters is an experimental test of their effect on the energy hierarchy and material cycles. The values of the parameters included in the program in Table 3 were those found by trial and error until all the graphs were semiquantitatively consistent with the definitions and theoretical relationships in Table 1. In other words, ideas of energy systems hierarchy were tested for self consistency with a synthetic simulation of the change of properties with steps in hierarchical level. This kind of simulation may be a new methodology for investigating systems hierarchy.

Table 2
Parameters for each level used in the program HIERSTEP.bas.
* indicates new properties to measure

Geometry and Time:

A	Whole area
at	Area of a territory
Ca	Fraction that the center occupies in its territory
ac	Area of all the centers at a level
aec	Area of one center at a level
sf	Fraction of area of centers at a lower level occupied by center at the next level*
nc	Number of centers at a level
N	Number of level 1 areas
Cr	Number of centers converging to a center at the nextlevel
R	Replacement time of a center storage
rf	Change in replacement time with hierarchical step*

Energy:

Je	Energy inflow for the whole area
eff	Efficiency of energy transformation in each step
ens	Energy stored in centers at a level

Materials:

M	Total materials
mc	Mass concentration at the low level
me	Materials transformed per unit available energy used (material coupling factor)*
meo	Initial material coupling factor*
fmc	Change in material coupling factor with level*
ms	Mass stored in centers at a level
mfld	Material flux density

Emergy:

emp	Empower for the whole area
tr	Transformity of flows
empd	Areal empower density
empg	Emergy per mass in flows
emter	Empower per territory
ems	Emergy stored in centers at a level
emg	Emergy per mass stored

Chapter 10. Simulating Energy and Materials in Hierarchical Steps




Figure 3. Properties generated by the program HIERSTEP.bas from the initial conditions assigned in lines 100-360 in Table 3.

Program HIERSTEP. bas

The program HIERSTEP.bas, with its statements, is listed in Table 3. The first part of the program (lines 10-350) has the initial conditions of the lowest level of hierarchy and scaling factors. There is one plot statement in line 420 plotting a variable Y (to be selected for each run) as a function of the level L. The lines 430-450 plot a bar according to the value of Y at the first level. Then the program steps to the next level (line 500) and reevaluates the variables (lines 510-700), returning on line 710 to the plot statement where the next bar is plotted, and so on until there is only one hierarchical center remaining. Then, with line 700, the program goes to the END in line 800.

The user types in the item to be plotted on the Y axis in the parenthesis in line 410. To represent the wide range of magnitudes in the different levels of hierarchy, plots are on logarithmic scale. In order to keep each item on the screen, screen factors were added by giving various plus or minus values to Y0, which causes the plot to shift up or down on the screen. Graphical outputs from 12 different runs were

Table 3

Listing of the BASIC program HIERSTEP.bas.

Load a form of basic such as QBASIC. Use its menu to open the program. Type in a property to be graphed in the parenthesis of Line 410 (example: (tr) to plot transformity. Use menu to select RUN. Or, print out values of a variable such as tr for levels by adding the following line: 402 PRINT tr: GOTO 500.

```

10 REM PC: HIERSTEP.bas (Properties of hierarchical steps)
15 CLS
20 SCREEN 1,0
30 COLOR 7,0
40 LINE (0, 0)-(200, 180), 3, B
45 S = 1:REM Step =1
50 L = 1:REM Start at level #1
80 REM Scaling Factors
83 DZ = 1: REM increment FOR counting bar WIDTH plot
84 DX = 1: REM Horizontal increment FOR plotting bar
86 X0 = 1: REM Factor for scaling horizontal axis
90 Y0 = 10: REM Factor for scaling vertical axis
100 REM First Level conditions and relationships
102 emp = 1E+20: REM Empower sej/time of the whole area A
105 M = 1E+07:REM Materials in lowest level
110 N = 1000: REM number of level 1 units and territories
120 Jeo = 1E+20:REM initial energy flow of the whole area
130 Je = Jeo:REM energy flow
140 eff = .1: REM fraction of energy transformed in each step
150 R = 1:REM initial replacement time
160 Rf = 10:REM factor changing replacment time with level
170 tr = emp/Je:REM initial transformity ofenergy flow of small areas
180 A = 1E+09:REM Whole area
185 mc = M/A:REM Mass concentration at the low level
190 at = A/N:REM area of each small territory
200 Ca = .05:REM Center fraction of territory area
210 aec = Ca*at:REM Area of each center
220 Sf = .3:REM area factor change per step

```

Table 3 HIERSTEP.bas (continued)

```

230 empd = emp/A:REM empower density of whole area
240 fmc = 2:REM factor changing material/energy coupling with level
250 meo = 1E-13:REM material coupling to energy flow
260 Jm = meo*mc*Je:REM flow of materials at the first level
270 ms = Jm*R:REM Mass stored
280 empg = emp/Jm:REM energy/mass in flows
290 Cr = 6:REM Number of centers converging to higher level center
300 nc = N:REM number of first level centers = number of territories
310 ac = aec*nc:REM area of the centers on the first level
320 mflld = Jm/ac:REM material flux density in centers of the first level
330 emter = emp/nc:REM empower per territory
340 ems = 1:REM emergy stored in background level
350 emg = empg/ms:REM emergy per mass stored
360 me = Jm/Je:REM materials transformed per unit energy, first level
400 REM Start of bar plotting or value printing Loop
410 Y = 10*LOG(empd)/2.31: REM Substitute variable here to be plotted
415 Y0 = 10
420 PSET (X/X0, 180 - (Y - Y0)), 3: REM Plotting whatever is in Line 410
430 Z = Z + DZ: REM bar width counter
440 X = X + DX
450 IF Z < 50 GOTO 400: REM
460 Z = 1
500 L = L + S:REM steps program to next energy transformation level
505 me = fmc*L*me:REM changed material/energy flow coupling
507 emp = emp:REM empower transfer
510 Je = Je*eff
520 Jm = me*Je:REM flow of materials at this level
530 at = at*Cr:REM area of each territory
550 tr = emp/Je
560 R = R*Rf:REM replacement time for this level
570 nc = A/at:REM Number of centers
580 ac = ac*Sf: REM Area of centers
590 aec = ac/nc: REM Area of each center
600 empd = emp/ac:REM empower density of centers
610 empg = emp/Jm:REM empower per mass flowing
620 mflld = Jm/ac:REM material flux density in centers
625 efld = Je/ac:REM energy flux density in centers
630 ms = Jm*R:REM Mass stored at the level
640 emter = emp/nc:REM empower per territory
650 mc = ms/ac:REM mass concentration
660 ems = emp*R:REM emergy stored at the level
670 emg = ems/ms:REM emergy per mass stored
680 ens = Je/R:REM energy stored
700 IF nc < 1 GOTO 800
710 IF X < 320 GOTO 400
800 END

```

assembled in Figure 2. Each was obtained by changing what was plotted in line 410 and its screen shifting factor in line 415. For example, to plot transformity of the flow for each level, type tr in line 410:

$$Y = 5 * \text{Log}(\text{Tr})$$

where 5 is a scaling factor for graphical presentation and Y0 is 10 shifting the plot up on the screen. When evaluated from real examples, the model is another way to generate hierarchical characteristics such as transformities, empower densities, energy per mass values plus the new parameters whose meaning may require more investigation.

REFERENCES

- Allen, T.F.H. and T.B. Starr. 1982. *Hierarchy, Perspectives in Ecological Complexity*, Univ. of Chicago Press, Chicago, IL.
- Berry, B.J.L., E.C. Conkling and D.M. Ray. 1976. *The Geography of Economic Systems*. Prentice-Hall, Englewood Cliffs, NJ, 529 pp.
- Clarke, G.L. 1959. Elements of Ecology. *Ecol. Monogr.* 16:321-335.
- Doxiadis, C.A. 1977. *Ecology and Ekistics*. Westview Press, Boulder, CO.
- Mandelbrot, B. 1982. *The Fractal Geometry of Nature*. Freeman, San Francisco, CA.
- Margalef, R. 1958. Temporal succession and spatial heterogeneity in phytoplankton. pp. 323-341 in *Perspectives in Marine Biology*, ed. by A.A. Buzaatti-Traverso. Univ. of Calif. Press, Berkeley, CA, 621 pp.
- Mesarovic, M.D., M. Mako, and Y. Takahara. 1970. *Theory of Hierarchical Multilevel Systems*. Academic Press, San Diego, CA.
- Odum, H.T. 1982. Pulsing, power and hierarchy. pp. 33-59 in *Energetics and Systems*, ed. by W.J. Mitsch, R.K. Ragade, R.W. Bosserman and J.A. Dillon, Jr. Ann Arbor Science, Ann Arbor, Mich.
- Odum, H.T. 1996. *Environmental Accounting, Energy and Decision Making*. John Wiley, NY, 373 pp.
- Odum, H.T. and E.C. Odum. 2000. *Modeling for All Scales, An Introduction to Simulation*. Academic Press, San Diego, 455 pp.
- Patee, H.H., ed. 1973. *Hierarchy Theory, the Challenge of Complex Systems*. Brazillier, NY, 158 pp.
- Steele, J. 1976. Application of theoretical models in ecology. *J. Theor. Biol.* 63:443-451.
- Weiss, P.B. 1973. *Hierarchy Organized Systems in Theory and Practice*. Hafner, NY.
- Whyte, L.L. and A.G. Wilson. 1969. *Hierarchical Structures*. Elsevier, NY.
- Zipf, G.K. 1941. *National Unity and Disunity*. Principia Press, Bloomington, IN, 407 pp.